

Setup

Data file: Chocolate Sales.csv

Disable Auto Date/Time in Power BI (File → Options → Current File → Data Load).

Create Date table:

```
Date =  
CALENDARAUTO()
```

```
Date[Year] = YEAR ( Date[Date] )  
Date[MonthNo] = MONTH ( Date[Date] )  
Date[MonthName] = FORMAT ( Date[Date], "MMM" )  
Date[YearMonth] = FORMAT ( Date[Date], "YYYY-MM" )
```

Mark the Date table as Date Table.

Relationship: choco_sales[Date] → Date[Date].

Base Measure
Revenue := SUM (choco_sales[Amount])

Basic Level (1–5)
-- 1. Total Sales (All-Time)
Total Sales (All-Time) := [Revenue]

-- 2. Current Year (YTD)
Sales CY (YTD) :=
TOTALYTD ([Revenue], 'Date'[Date])

-- 3. Last Year (YTD)
Sales LY (YTD) :=
CALCULATE ([Revenue], PREVIOUSYEAR ('Date'[Date]))

-- 4. Current Month (MTD)
Sales CM (MTD) :=
TOTALMTD ([Revenue], 'Date'[Date])

-- 5. Current Quarter (QTD)
Sales CQ (QTD) :=
TOTALQTD ([Revenue], 'Date'[Date])

Intermediate Level (6–10)

-- 6. Year-over-Year % Growth

Sales LY (context) :=

CALCULATE ([Revenue], SAMEPERIODLASTYEAR ('Date'[Date]))

YoY % :=

VAR LY = [Sales LY (context)]

RETURN DIVIDE ([Revenue] - LY, LY)

-- 7. Sales from Last Month

Sales LM :=

CALCULATE ([Revenue], DATEADD ('Date'[Date], -1, MONTH))

-- 8. Running Total

Running Total :=

CALCULATE ([Revenue],
FILTER (ALL ('Date'[Date]), 'Date'[Date] <= MAX ('Date'[Date])))

-- 9. Sales for Last 3 Months

Sales L3M :=

CALCULATE ([Revenue],
DATESINPERIOD ('Date'[Date], MAX ('Date'[Date]), -3, MONTH))

-- 10. Month with Highest Sales in Last 12 Months

Top Month L12M (Amount) :=

VAR Win =

DATESINPERIOD ('Date'[Date], MAX ('Date'[Date]), -12, MONTH)

VAR T =

SUMMARIZE (Win,
YEAR ('Date'[Date]), MONTH ('Date'[Date]),
"Amt", CALCULATE ([Revenue]))

RETURN MAXX (TOPN (1, T, [Amt], DESC), [Amt])

Top Month L12M (Label) :=

VAR Win =

DATESINPERIOD ('Date'[Date], MAX ('Date'[Date]), -12, MONTH)

VAR T =

ADDCOLUMNS (SUMMARIZE (Win, 'Date'[Date]),

```

    "Amt", CALCULATE ( [Revenue] ),
    "Label", FORMAT ( 'Date'[Date], "YYYY-MM" )
)
RETURN MAXX ( TOPN ( 1, T, [Amt], DESC ), [Label] )

```

Advanced Level (11–15)

-- 11. Q1 Sales of Each Year

Q1 Sales :=

```

CALCULATE (
    [Revenue],
    FILTER ( VALUES ( 'Date'[Date] ), QUARTER ( 'Date'[Date] ) = 1 )
)

```

-- 12. YoY Difference Only for December

Dec YoY Difference :=

VAR IsDec = MONTH (MAX ('Date'[Date])) = 12

VAR Cur = [Revenue]

VAR Prev = CALCULATE ([Revenue], SAMEPERIODLASTYEAR ('Date'[Date]))

RETURN IF (IsDec, Cur - Prev)

-- 13. Rolling 12 Months Sales

Sales L12M :=

```

CALCULATE (
    [Revenue],
    DATESINPERIOD ( 'Date'[Date], MAX ( 'Date'[Date] ), -12, MONTH )
)

```

-- 14. Difference Between Current and Previous Quarter

Sales QTD := TOTALQTD ([Revenue], 'Date'[Date])

Sales QTD (Prev) :=

CALCULATE ([Sales QTD], DATEADD ('Date'[Date], -1, QUARTER))

QoQ Difference := [Sales QTD] - [Sales QTD (Prev)]

-- 15. Flag: Months > 110% of Last Year

YoY 110%+ Flag :=

VAR LY = CALCULATE ([Revenue], SAMEPERIODLASTYEAR ('Date'[Date]))

RETURN IF (NOT ISBLANK (LY) && [Revenue] > 1.1 * LY, 1, 0)

 Tips for visualization:

Use Year, Month, or YearMonth from the Date table on the X-axis.

For rolling totals (3M, 12M), use a line chart with continuous date.

For Q1 comparison, put Year on rows and Q1 Sales as values.

For YoY 110%+ Flag, apply conditional formatting in a matrix or bar chart (1 = highlight).